

ТЕМА 3

Учимся работать с символами

В предыдущей теме мы рассмотрели типы данных, позволяющие хранить и обрабатывать числа — целые и дробные. Но, перефразируя известную поговорку, не числами едиными живет программист. Кроме чисел, Паскаль умеет также работать с символьной информацией. Для каждого символа в программе выделяется отдельная ячейка со всеми присущими ячейке параметрами — именем и значением.

Урок 3.1. Как компьютер понимает символы

Под символами мы понимаем буквы и все те значки, которые вы видите на клавиатуре. В Паскале переменные для хранения символов имеют тип `Char`.

За каждым символом закреплен свой числовой код. Все коды сведены в таблицу.

Кодовая таблица ASCII

Обычно для хранения символов используют код, называемый ASCII (американский стандартный код обмена информацией).

Как видите, цифры здесь — не числовые данные, а тоже символы, каждый из которых имеет свой код. В компьютере коды записаны в двоичном виде. На каждый код выделено 8 бит, то есть 1 байт. Получаем $2^8 = 256$ двоичных кодов. Всего в таблице ASCII 256 кодов: наименьшее значение кода 00000000, наибольшее — 11111111 (это 255 в двоичном виде).

Таблица 3.1. Фрагмент таблицы ASCII (таблица кодов символов)

Код	Двоичный код	Символ	Код	Двоичный код	Символ
48	00110000	0	65	01000001	A
49	00110001	1	66	01000010	B
50	00110010	2	67	01000011	C
57	00111001	9	89	01011001	Y
			90	01011010	Z

Описание типа Char и стандартные функции

Пример 3.1. Демонстрация стандартных функций для работы с типом Char

```

Program Letter1;
var
  N: Integer;
  X: Char;
begin
  X:='L'; { В символьную переменную X
           записали символ 'L' }

  writeln(X);

  N:=Ord (X); { Функция Ord возвращает код символа,
               занесенного в переменную X, то есть код
               буквы 'L' }

  writeln(N);
  X:='A';
  writeln(X);

  X:=Chr (N); { Функция Chr возвращает символ
               по заданному коду. Сейчас в переменной X
               оказался символ 'L' – именно его код мы
               только что записали в переменную N }

  writeln(X);
  readln
end.

```

При выполнении программа выведет на экран следующее:

```
L
76
A
L
```

Пример 3.2. Ввод символов с клавиатуры

```
Program Letter2;
var
  X,Y:Char;
begin
  writeln('Введите несколько символов:');
  readln(X);
  writeln(X);
  writeln('Введите еще несколько символов:');
  readln(X,Y);
  writeln(X,Y);
  readln
end.
```

Запустив программу на выполнение, введите с клавиатуры последовательность символов (например, ABC) и нажмите Enter. Программа выведет единственный символ:

```
A
```

В ответ на второе предложение введите с клавиатуры CAT. На экране получим следующее:

```
CA
```



ЗАМЕЧАНИЕ

Переменная типа Char принимает только один символ из введенной строки. При вводе символы не заключаются в апострофы. Таким образом, в первом случае из введенных символов запомнился только один, во втором — два.

Можно определять и символьные константы:

```
const Leto='X';
```

Урок 3.2. Тип Char — порядковый тип!

В таблице кодов вы могли заметить такую закономерность:

'0' < '1' < '2' < '3' < ... < '9' < ... 'A' < 'B' < 'C' < ... < 'X' < 'Y' < 'Z' ... (коды символов упорядочены).

Таким образом, для каждого элемента типа Char всегда есть предшествующий и последующий элементы. Такой тип данных называется *порядковым*. Тип Char — порядковый тип. Тип Integer также является порядковым.

Пример 3.3. Стандартные функции, применяемые к порядковому типу

```

Program Letter3;
var
  X1,X2,X3,X4: Char;
begin
  X1:='L';
  writeln(X1);

  X2:=Pred (X1); { Функция Pred возвращает
                  предшествующий элемент
                  относительно значения
                  переменной X1 }

  writeln('Pred=',X2);

  X3:=Succ (X1); { Функция Succ возвращает
                  последующий элемент
                  относительно значения
                  переменной X1 }

  writeln('Succ=',X3);
  readln
end.
```

При выполнении программа выведет на экран следующее:

```

L
Pred=K
Succ=M
```

Задание 3.1. Напишите программу расшифровки 4-буквенного однословного сообщения. Для получения 4 букв нужно ввести 3 строки:

- ✦ из первой строки прочесть только первую букву;
- ✦ из второй строки прочесть только первую букву;
- ✦ из третьей строки прочесть первую и вторую буквы.

Далее расшифровать полученные четыре буквы по такому алгоритму: вместо первой и третьей букв подставить соответственно буквы, отстоящие от них по алфавиту на две буквы назад, а вторую и четвертую буквы оставить без изменения.

Для проверки возьмите пример, приведенный ниже.

Ввод:

```
FINISHED
OR
PENDING?
```

На выводе должно быть слово «DONE».

Задание 3.2. Известно, что коды прописных (заглавных) букв латинского алфавита следуют в таблице непрерывно друг за другом. Коды строчных букв латиницы также следуют непрерывно друг за другом на расстоянии 32 символов от прописных (ниже по таблице). Если $\text{ord}('A') = 65$, то $\text{ord}('A') + 32 = 97$, и это код строчной буквы «а», то есть $\text{chr}(\text{ord}('A') + 32) = 'a'$. Напишите программу, в которой вы вводите прописную букву (только латиницу!), а получаете ее строчный эквивалент, и наоборот, по строчной букве получаете соответствующую прописную.



ЗАМЕЧАНИЕ

С русскими символами такого порядка нет из-за особенностей организации кодовой таблицы. В частности, строчные буквы в таблице следуют не подряд, а с разрывом в середине алфавита.

Выводы

1. Все символы хранятся в компьютере в виде кодов.
2. Обычно для кодирования символов применяется таблица ASCII.
3. Каждому символу соответствует свой код.
4. Для преобразования символов в коды и обратно применяют функции `ord` и `chr`.
5. Тип `Char` является порядковым типом.

6. Коды буквы латинского алфавита идут последовательно.
7. Русские буквы хранятся в таблице символов ASCII с разрывом в последовательности кодов.
8. Для получения следующего и предыдущего символа используют соответственно функции `succ` и `pred`.

Контрольные вопросы

1. Сколько всего различных символов кодируется таблицей ASCII?
2. Какой объем памяти требуется для кодирования одного символа? А для 15 символов?
3. Какой тип данных в Паскале предназначен для хранения символьной информации? Сколько символов можно поместить в одну переменную этого типа?
4. Какой код у буквы «F»? Какой символ кодируется кодом 87?
5. В программе определены 3 переменные (`a, b, c: char;`). В ответ на инструкцию `readln(b, a, c)`; пользователь ввел текст Леша. В каком месте памяти оказалась каждая из введенных букв?
6. Каков будет результат выполнения инструкции `c:=succ(pred(succ('D')))`?
7. Какое значение получит переменная `i` в операторе `i := pred(ord('F')-2)`?